# Maintainer: Alexey Pavlov <alexpux@gmail.com>

_realname=TimeDate
pkgname="perl-${_realname}"
pkgver=2.31
pkgrel=1
pkgdesc="Date formating subroutines"
arch=('any')
license=('PerlArtistic')
url="https://metacpan.org/${_realname}/"
depends=('perl')
options=('!emptydirs')
source=("https://www.cpan.org/authors/id/A/AT/ATOOMIC/${_realname}-${pkgver}.tar.gz")
sha256sums=('5c720fedb245122d073ea9c030aca24b06a615c71d40c46f832a8a1809354d81')

build() {
  cd ${_realname}-${pkgver}

  # install module in vendor directories.
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make install DESTDIR="${pkgdir}"
}
